Custom Deep Copies


by Paul Baumann





This article provides a reusable framework to deep copy to any desired depth. The desired copy depth can depend not only on what the object is, but also on how it is being used. It can copy the same object differently depending on the context in which the copy is made, and the context will be maintained throughout the entire deep copy session.


Deep Copy Issues


Smalltalk implementations generally provide a copy method that simply creates a shallow copy of an object. A shallowCopy only creates a copy of an object that contains the same contents as the original object. VisualAge slightly improves on this by implementing a deepCopy method which makes a copy of the receiver object and shallow copies of its contents. Unfortunately, deepCopy doesn’t copy deep enough for many purposes. To create truly deep copies, VisualAge also implements an abtClone method. While the abtClone method fulfills the need for generically creating truly deep copies of an object, it isn’t suited for creating customized deep copies that preserve the context in which the object is being copied.





A custom deep copy is most applicable to objects that have complex relationships with other objects. When you copy these objects, you want the copy to maintain the same relationships with referenced objects as the original. For example, you wouldn’t want a backpointer held in a copy of a referenced object to point back to the original object that was copied; but rather, you would want the backpointer to point to the new copy of the original object. If a deep copy mechanism were to ignore the relationships between objects, then any backpointers encountered would start a copy loop that would end with a stack overflow error. It is because of looping recursion problems that the standard VisualAge deepCopy is as shallow as it is.





The key to making deep copies that maintain recursive references is to copy an object in two stages. In the first stage, only the structure is copied and added to a collection of copied objects. To avoid recursion problems, the structure of the object is used as a place holder that other copied objects can refer to before the object has finished being copied. In the second stage, the new structure is given the opportunity to populate itself from the original object. The copy can’t be performed all at once because an object’s contents may refer back to the object that is still in the process of being copied, but the new copy of the object can’t be referenced because it hasn’t yet been copied. 





Other implementations differ in that they don’t separate the logic to control the copy from the methods that declare what and how to copy. In doing so, they forgo having a layered approach that allows for more than one way to copy an object. Other implementations can copy contextually while still reusing some code, but only at the expense of losing the original copy context for all remaining elements to be copied.





This code is unlike other deep copy implementations because it applies a Strategy� pattern to control the copying of each object. The strategy contains all behavior that is common when copying an object, and is used when copying all remaining elements. It is used to hold data related to a copy session, to control the sequence of copy operations, and to handle special circumstances. It is the best way to provide a layered true deep copy that can handle recursion.


A Layered Implementation


A layered approach is used to allow for easy customization with minimal effort. Method-based inheritance is used to define each layer. Each layer has methods that unless overridden, just forward copy commands to the next lowest layer. The typical copy only uses the first three layers which provide support for copies in general. Layers four and higher are developer defined to copy objects in a non-standard way for a specific context.





The first layer is the basic layer and is shown with the base code in Figure 1. Methods of this layer should rarely be overridden. It is only separated from the other layers so that any higher-layer overrides in a superclass can be undone for any subclasses that should use the now overridden defaults. If not overridden by any of the other layers, then an object will be copied using any code in the basic layer.





The second layer is the vendor specific layer and is shown in Figure 2. Its default behavior for most classes is to just forward copy instructions to the basic layer. Code in this layer should be specified by the Smalltalk vendor to declare how a particular Smalltalk version should, for example, copy a collection. Immutable objects (like nil, integers, booleans, characters, symbols, atoms, etc.) also require an override in the vendor layer to avoid an attempt to copy instances. 





Only classes defined by a vendor should implement overrides in this layer. Because these methods aren’t already specified by vendors, you may have to determine what some of the methods should be yourself. Fortunately, you can use the vendors implementations of copy, shallowCopy, postCopy, deepCopy, abtClone, etc. as a guide. Also, most of the required code for VisualAge 4.0 and VisualWorks 2.5.1 is provided for you in this article. You might also find updates and ports of the code online�.





The third layer is the general layer and is shown in Figure 3. Its default behavior is to just forward copy instructions to the vendor layer. It is in this layer that you declare how an object will be copied for most circumstances. This layer is concerned with how instances of classes that you defined should be copied. 





Layers four and higher contain developer defined code to copy objects for a specific context. You can create as may layers as you need, and can inherit from any of the layers. You can also reuse code from any of the other layers without forgetting the original context layer used to start the copy.


Example Code


If you needed to copy instances in a way that excludes any confidential information then you could add methods to define a confidentialCopy layer. Lets assume that an Employee instance refers to a collection of instances of Address as well as a collection of instances of EmployeeReview. Lets also assume that a confidentialCopy shouldn’t include information such as salary or review information. Also, only city and state can be disclosed by copies of address instances.





We will define the confidentialCopy layer to sit on top of the general layer. That way we can add override methods to only a few classes and still reuse code from any of the other layers. A minimum of three methods must be implemented on Object whenever a new layer is defined.  Here are the three methods we will use to define the base for our confidentialCopy layer example:


Object>>confidentailCopy


"Answer a true deep copy of the receiver that doesn’t include any confidential information."


	| copyStrategy |


	^(copyStrategy := PlbCopyStrategy new)


		structureCopyBlock: [:original | 


			original confidentialCopyStructure 


		];


		contentsCopyBlock: [:original :skeleton | 


			skeleton 


				confidentialCopyContentsFrom: original 


				using: copyStrategy 


		];


		copyRoot: self.


Object>>confidentialCopyStructure


"Answer an unpopulated instance like the receiver. This skeleton instance will later be populated as each referenced variable of the receiver is copied. Answer nil or the receiver if the receiver shouldn't be copied."


	^self plbGeneralCopyStructure


Object>>confidentialCopyContentsFrom: original using: copyStrategy


"Populate the variable slots of the receiver with copies of each variable slot of original created under the control of copyStrategy."


	self plbGeneralCopyContentsFrom: original using: copyStrategy





With the base defined for our new layer we can now ask any object for a confidential copy. But with only the above three methods defined, a confidentialCopy is still no different from a plbGeneralCopy. We also need to add methods to differentiate this layer from the general layer:


Employee>>confidentialCopyContentsFrom: original using: copyStrategy


"Copy all variables then nil out the salary."


	super confidentialCopyContentsFrom: original using: copyStrategy.


	salary := nil.


Address>>confidentialCopyContentsFrom: original using: copyStrategy


"All variables except for city and state are considered confidential."


	city := copyStrategy copy: original city.


	state := copyStreategy copy: original state.


EmployeeReview>>confidentialCopyStructure


"Answer nil since instances of EmployeeReview shouldn’t be copied."


	^nil





Notice how simple the overrides are to customize the confidentailCopy layer. Because this is a layered approach, you only have to implement changes from the inherited layer. You don’t need to know and maintain intimate details about object structure or declare how collections should copy themselves. Nor do you need to create or duplicate copy methods just to pass along the context of the copy. The code maximizes reuse and maintainability allowing you to concentrate on the task at hand.


Additional Notes


It should be noted that because Smalltalk supports blocks, we were able to specify how the copy strategy is to operate—from outside of the code that defines the strategy. It is the plbGeneralCopy method that specifies what is to be copied—not the strategy. The result is easily customizable code. Without blocks, or at least an implementation of a Command� pattern, we would have to create subclasses of the copy strategy class in order to perform customized copies. It is yet another example of how blocks and class extensions give Smalltalk an advantage over languages like Java.


Conclusion


Many projects use a coding style that involves copying objects quite frequently. If your project copies complex objects frequently, or for multiple purposes, then this copy framework can be used to simplify and standardize how copies are made. 





Because of the layered implementation, the copy code you write is not tied to any particular Smalltalk vendor’s code. This improves the portability of your code by allowing you to ignore base class implementations—provided a vendor layer is already defined. It is hoped that vendors will incorporate this layered deep copy technique in their future product releases.





Paul Baumann has been using Smalltalk since 1993. He is an IBM certified VisualAge for Smalltalk Developer, and can be reached at pbaumann@ibm.net.





�
Figure 1: (the Base Code and Basic Layer) 


Object subclass: #PlbCopyStrategy


    instanceVariableNames: 'visited structureCopyBlock contentsCopyBlock '


    classVariableNames: ''


    poolDictionaries: ''


PlbCopyStrategy class>>new


"Answer a new instance of the receiver with a number of preallocated slots for unique objects."


	^self new: 32


PlbCopyStrategy class>>new: uniqueObjectEstimate


"Answer a new instance of the receiver with a number of preallocated slots for unique objects."


	^self basicNew initialize: uniqueObjectEstimate


PlbCopyStrategy>>initialize: uniqueObjectEstimate


"Set visited to a collection that can be used to identify objects that were already copied."


	visited := IdentityDictionary new: uniqueObjectEstimate.


PlbCopyStrategy>>structureCopyBlock: oneArgBlock


"Set an one argument block to use for copying the structure of each object. The first argument will be the original object to copy."


	structureCopyBlock := oneArgBlock


PlbCopyStrategy>>contentsCopyBlock: twoArgBlock


"Set a two argument block to use for copying the contents of each object. The first argument is the original object. The second argument is a new empty object like the original."


	contentsCopyBlock := twoArgBlock


PlbCopyStrategy>>copyRoot: anObject


“Answer a copy of anObject. This method differs from the #copy: method in that it can be overridden for special processing on the first object to be copied.”


	^self copy: anObject


PlbCopyStrategy>>copy: anObject


“Answer a copy of anObject as controlled by the receiver.”


	^visited 


		at: anObject 


		ifAbsent: [ self createCopyOf: anObject ].


PlbCopyStrategy>>createCopyOf: anObject


"Answer a copy of anObject after adding it to a collection of copied objects."


	| copy | 


	copy := visited


		at: anObject


		put: (structureCopyBlock value: anObject).


	copy == anObject ifTrue: [ ^anObject ].


	copy isNil ifFalse: [ 


		contentsCopyBlock 


			value: anObject 


			value: copy 


	].


	^copy


The Basic Layer:


Object>>plbBasicCopyStructure


"Answer an unpopulated instance like the receiver. This skeleton instance will later be populated as each referenced variable of the receiver is copied. Answer nil or the receiver if the receiver shouldn't be copied."


	^self class isVariable


		ifTrue: [ self class basicNew: self basicSize ]


		ifFalse: [ self class basicNew ]


Object>> plbBasicCopyContentsFrom: original using: copyStrategy


"Populate the variable slots of the receiver with copies of each variable slot of original created under the control of copyStrategy."


	1 to: self class instSize do: [ :instVarIndex |


		self


			instVarAt: instVarIndex


			put: (copyStrategy copy: (original instVarAt: instVarIndex))


	].


	1 to: self basicSize do: [ :index |


		self 


			basicAt: index 


			put: (copyStrategy copy: (original basicAt: index))


	].


Figure 2: (the Vendor Layer) 


Object>>plbVendorCopy


"Answer a true deep copy of the receiver while maintaining references. Each referenced object will have the opportunity to participate in the copy and decide how it should be copied."


	| copyStrategy |


	^(copyStrategy := PlbCopyStrategy new)


		structureCopyBlock: [:original | 


			original plbVendorCopyStructure 


		];


		contentsCopyBlock: [:original :skeleton | 


			skeleton 


				plbVendorCopyContentsFrom: original 


				using: copyStrategy 


		];


		copyRoot: self.


Object>>plbVendorCopyStructure


"Answer an unpopulated instance like the receiver. This skeleton instance will later be populated as each referenced variable of the receiver is copied. Answer nil or the receiver if the receiver shouldn't be copied."


	^self plbBasicCopyStructure


Object>>plbVendorCopyContentsFrom: original using: copyStrategy


"Populate the variable slots of the receiver with copies of each variable slot of original created under the control of copyStrategy."


	self plbBasicCopyContentsFrom: original using: copyStrategy


Boolean>>plbVendorCopyStructure


"Booleans shouldn't be copied. Just answer the receiver."


	^self


Character>>plbVendorCopyStructure


"Characters shouldn't be copied. Just answer the receiver."


	^self


Collection>>plbVendorCopyStructure


"Answer an unpopulated instance like the receiver. This skeleton instance will later be populated as each referenced variable of the receiver is copied."


	^self class new: self size


Collection>>plbVendorCopyContentsFrom: original using: copyStrategy


"Populate the receiver with copies of each element of original created under the control of copyStrategy."


	original do: [:item| self add: (copyStrategy copy: item)].


Interval>>plbVendorCopyStructure


"Answer a standard copy of the receiver."


	^self copy


Interval>>plbVendorCopyContentsFrom: original using: copyStrategy


"Do nothing because the structure copy already copied the contents."


Number>>plbVendorCopyStructure


"Answer a standard copy of the receiver."


	^self copy


Number>>plbVendorCopyContentsFrom: original using: copyStrategy


"This method will only be run if a subclass of Number is created that does not answer self when asked for a copy. The situation is unlikely."


SortedCollection>>plbVendorCopyStructure


"Answer an unpopulated instance using the same sort block as the receiver."


	^self class sortBlock: self sortBlock


Symbol>>plbVendorCopyStructure


"Symbols shouldn't be copied. Just answer the receiver."


	^self


UndefinedObject>>plbVendorCopyStructure


"nil shouldn't be copied. Just answer the receiver."


	^self


Vendor Code for VisualAge 4.0 only:


ArrayedCollection>>plbVendorCopyContentsFrom: original using: copyStrategy


"Populate the variable slots of the receiver with copies of each variable slot of original created under the control of copyStrategy."


	original doWithIndex: [:item :index |


		self at: index put: (copyStrategy copy: item)


	].


EsAtom>>plbVendorCopyStructure


"Atoms shouldn't be copied. Just answer the receiver."


	^self


EsString>>plbVendorCopyStructure


"Answer a standard copy of the receiver."


	^self copy


EsString>>plbVendorCopyContentsFrom: original using: copyStrategy


"Do nothing because the string has already been copied."


KeyedCollection>>plbVendorCopyContentsFrom: original using: copyStrategy


"Populate the receiver with copies of each element of original created under the control of copyStrategy."


	original associationsDo: [:assoc | 


		self 


			at: assoc key


			put: (copyStrategy copy: assoc value)


	].


Vendor Code for VisualWorks 2.5.1 only:


ArrayedCollection>>plbVendorCopyContentsFrom: original using: copyStrategy


"Populate the variable slots of the receiver with copies of each variable slot of original created under the control of copyStrategy."


	| item |


	1 to: original size do: [:index |


		item := original at: index.


		self at: index put: (copyStrategy copy: item)


	].


Dictionary>>plbVendorCopyContentsFrom: original using: copyStrategy


"Populate the receiver with copies of each element of original created under the control of copyStrategy."


	original associationsDo: [:assoc | 


		self 


			at: assoc key


			put: (copyStrategy copy: assoc value)


	].


String>>plbVendorCopyStructure


"Answer a standard copy of the receiver."


	^self copy


String>>plbVendorCopyContentsFrom: original using: copyStrategy


"Do nothing because the string has already been copied."


VariableBinding>>plbVendorCopyStructure


"VariableBindings shouldn't be copied. Just answer the receiver."


	^self


Figure 3: (the General Layer)


Object>>plbGeneralCopy


"Answer a true deep copy of the receiver while maintaining references. Each referenced object will have the opportunity to participate in the copy and decide how it should be copied."


	| copyStrategy |


	^(copyStrategy := PlbCopyStrategy new)


		structureCopyBlock: [:original | 


			original plbGeneralCopyStructure 


		];


		contentsCopyBlock: [:original :skeleton | 


			skeleton 


				plbGeneralCopyContentsFrom: original 


				using: copyStrategy 


		];


		copyRoot: self.


Object>>plbGeneralCopyStructure


"Answer an unpopulated instance like the receiver. This skeleton instance will later be populated as each referenced variable of the receiver is copied. Answer nil or the receiver if the receiver shouldn't be copied."


	^self plbVendorCopyStructure


Object>>plbGeneralCopyContentsFrom: original using: copyStrategy


"Populate the variable slots of the receiver with copies of each variable slot of original created under the control of copyStrategy."


	self plbVendorCopyContentsFrom: original using: copyStrategy


Behavior>>plbGeneralCopyStructure


"Behaviors shouldn't generally be copied. Answer the receiver."


	^self


Context>>plbGeneralCopyStructure


"Contexts shouldn't generally be copied. Answer the receiver."


	^self


Number>>plbGeneralCopyStructure


"Answer the receiver because operations on numbers always create new numbers."


	^self


String>>plbGeneralCopyStructure


"Since strings are rarely modified by #at:put:, the general behavior is to just answer the receiver."


	^self


General Code for VisualAge 4.0 only:


CompiledMethod>>plbGeneralCopyStructure


"Compiled code shouldn't generally be copied. Answer the receiver."


	^self


EsPoolDictionary>>plbGeneralCopyStructure


"PoolDictionaries shouldn't generally be copied. Answer the receiver."


	^self


EsSmalltalkDictionary>>plbGeneralCopyStructure


"The system dictionary shouldn't generally be copied. Answer the receiver."


	^self


EsSmalltalkNamespace >>plbGeneralCopyStructure


"The system dictionary shouldn't generally be copied. Answer the receiver."


	^self


General Code for VisualWorks 2.5.1 only:


CompiledCode>>plbGeneralCopyStructure


"Compiled code shouldn't generally be copied. Answer the receiver."


	^self


PoolDictionary>>plbGeneralCopyStructure


"PoolDictionaries shouldn't generally be copied. Answer the receiver."


	^self


SystemDictionary>>plbGeneralCopyStructure


"The system dictionary shouldn't generally be copied. Answer the receiver."


	^self





� Design Patterns: Elements of Reusable Object-Oriented Software, Erich Gamma…[et al.]. pp. 315-323.


� Look for code in the UIUC Smalltalk Archive at http://st-www.cs.uiuc.edu/. If not available there then request code from the author by emailing pbauman@ibm.net.


� Design Patterns: Elements of Reusable Object-Oriented Software, Erich Gamma…[et al.]. pp. 233-242.











